# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=gtk-theme-elementary
pkgver=7.3.0
pkgrel=1
pkgdesc='elementary GTK theme'
arch=(any)
url=https://github.com/elementary/stylesheet
license=(GPL3)
groups=(pantheon)
makedepends=(
  git
  meson
  sassc
)
optdepends=('gtk-engine-murrine: GTK2 theme')
_tag=d521bf955423330ac809616d935b9063cbadd5cb
source=(gtk-theme-elementary::git+https://github.com/elementary/stylesheet.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd gtk-theme-elementary
  git describe --tags
}

build() {
  arch-meson gtk-theme-elementary build
  ninja -C build
}

package() {
  DESTDIR="${pkgdir}" ninja -C build install
}

# vim: ts=2 sw=2 et:
